
1 (6 points) Put an X in the True or False box for each statement.

(a) If the goal in the current proof state is 3 <= 3, then using the tactic reflexivity will solve
the goal.

□ True ⊠ False

(In the part of Rocq that we’ve seen in class, it will not, and so we took False as the correct
answer. But importing certain standard Rocq libraries (e.g., List) will actually make it work!)

(b) For any f, if H : f x = f y is an assumption in the current context, then inversion H will
discover that x = y.

□ True ⊠ False

(c) If X is an inductively defined type or proposition with no constructors and foo : X, then
destruct foo will finish any proof or subgoal.

⊠ True □ False

(d) The function fun (X : Type) (x : X) (f : X -> X) => f x terminates on all inputs.

⊠ True □ False

(e) A boolean function f : nat -> bool reflects a property P of numbers (P : nat -> Prop)
exactly when forall (n:nat), (f n = true) <-> P n.

⊠ True □ False

(f) If E has type beq_nat m n = true, then E also has type m = n.

□ True ⊠ False

2 (14 points) Write the type of each of the following Rocq expressions (or write “ill-typed” if
an expression does not have a type).

(a) 4 <= 3

Answer: Prop

(b) forall (A : Type) (m n : A), m = n \/ m <> n

Answer: Prop

(c) fun (x : nat) => False

Answer: nat -> Prop

(d) forall (m : nat), m * m

Answer: ill-typed

(e) beq_nat 3

Answer: nat -> bool

(f) fun (P Q : Prop) => P -> Q

Answer: Prop -> Prop -> Prop

(g) fun (m : nat) (E : 0 <= m) => le_S 0 m E

Answer: forall (m:nat), 0 <= m -> 0 <= S m

3 (16 points) For each of the types below, write a Rocq expression that has that type or write
“Empty” if there are no such expressions.

You can use global definitions defined in the appendix in your answer (e.g., the expression
fun (X Y : Type) => list (X * Y) would be a valid answer for the type Type -> Type -> Type)
but no other ones (e.g., option from Rocq’s standard library cannot be used).

(a) nat -> nat -> bool

Answer: Example: leb

(b) forall (X Y : Type), list X -> list Y

Answer: Example: fun X Y (l : list X) => []

(c) forall (X Y : Type) (f : X -> Y), Y

Answer: Empty

(d) Prop -> bool

Answer: Example: fun (P : Prop) => true

(e) In 2 [1;1;1]

Answer: Empty

(f) ev 1

Answer: Empty

(g) forall n : nat, ev n -> ev (S (S n))

Answer: Example: ev_SS

(h) (nat -> nat) -> nat

Answer: Example: fun f => (f 0)

4 This problem asks you to translate mathematical ideas from English into Rocq.

(a) (5 points) The concept of prefix of a list can be defined as an inductive relation as follows:

Inductive prefix {X : Type} : list X -> list X -> Prop :=
| prefix_nil : forall l, prefix [] l
| prefix_cons : forall t l h, prefix t l -> prefix (h::t) (h::l).

For example, these are all the prefixes of [1;2;3] (i.e., all the lists ls such that prefix ls [1;2;3]
is provable):

[]
[1]
[1;2]
[1;2;3]

Conversely, a suffix of a list is a substring that occurs at the end of the larger string. For
example, here are all the suffixes of [1;2;3]:

[]
[3]
[2;3]
[1;2;3]

Complete the following inductive definition of the suffix relation, where suffix s l indicates
that s is a suffix of l.

Inductive suffix {A : Type} : list A -> list A -> Prop :=

Answer:

| suffix_id : forall l, suffix l l
| suffix_tail : forall s t h, suffix s t -> suffix s (h::t).

(b) (8 points) A binary tree with labels of type A is either empty or a branch that has some value
of type A along with two binary trees of type A as children. Its formal definition is provided
on page ?? of the appendix. For example, the definition

Example exTree : tree nat :=
Branch 5

(Branch 2
(Branch 1 Empty Empty)
(Branch 4 Empty Empty))

(Branch 9
Empty
(Branch 7 Empty Empty)).

represents the following binary tree:

A subtree of a binary tree is either the tree itself or a subtree of one of its children. For
example, here are all the subtrees of exTree:

Empty 1 4 7

2

1 4

9

7

5

2

1 4

9

7

Give an inductive defininition of the subtree relation on binary trees, where subtree s t
indicates that s is a subtree of t.

Inductive subtree {A : Type} : tree A -> tree A -> Prop :=

Answer:

| subtree_id : forall t, subtree t t
| subtree_lt : forall s l a r, subtree s l -> subtree s (Branch a l r)
| subtree_rt : forall s r a l, subtree s r -> subtree s (Branch a l r).

5 (12 points) This problem asks you to translate mathematical ideas from Rocq into English.
In each part, your task is to find a short English summary of the meaning of a proposition defined
in Rocq. For example, if we gave you this definition...

Inductive D : nat -> nat -> Prop :=
| D1 : forall n, D n 0
| D2 : forall n m, (D n m) -> (D n (n + m)).

... your summary could be “D m n holds when... m divides n with no remainder.”

(a) Inductive R {X : Type} : X -> list X -> Prop :=
| R1 : forall x l, R x (x::l)
| R2 : forall x y l, (R x l) -> (R x (y::l)).

R X x l holds when... Answer: x occurs in the list l

(b) Inductive R {X : Type} : list X -> list X -> Prop :=
| R1 : R [] []
| R2 : forall x l1 l2, (R l1 l2) -> (R l1 (x::l2))
| R3 : forall x l1 l2, (R l1 l2) -> (R (x::l1) (x::l2)).

R X l1 l2 holds when... Answer: l1 is a subsequence of l2

(c) Inductive R {X : Type} : list X -> list X -> Prop :=
| R1 : R [] []
| R2 : forall x l1 l2 l3 l4,

(R (l1++l2) (l3++l4)) ->
(R (l1++[x]++l2) (l3++[x]++l4)).

R X l1 l2 holds when... Answer: l1 is a permutation of l2

(d) Definition R (m : nat) :=
m > 1 /\ (forall n, 1 < n -> n < m -> ~(D n m)).

(where D is given at the top of the page).

R m holds when... Answer: m is prime

6 (16 points) In this problem you will help write a simple regular expression matching function,
re_match. The expression re_match s re should yield true if and only if the “string” s (a list of
nats) matches the regular expression re. For example:

re_match [1; 3] (App (Union (Char 1) (Char 2)) (Char 3)) = true
re_match [1; 2; 3] (App (Union (Char 1) (Char 2)) (Char 3)) = false

Complete the recursive definition of the function re_match on the following page, filling in the
branches for EmptySet, Char, App and Union. Your answers may use any function defined in the
appendix. In particular, the following functions might be useful:

• list_nat_eq : list nat -> list nat -> bool
checks whether two lists are equal.

• existsb : forall X : Type, (X -> bool) -> list X -> bool
checks whether at least one element in a list l : list X satisfies a boolean property
p : X -> bool.

• forallb : forall X : Type, (X -> bool) -> list X -> bool
checks whether all elements of a list l : list X satisfy a boolean property p : X -> bool.

• splits : forall X : Type, list X -> list (list X * list X)
enumerates all possible ways to split a list s into two lists (s1, s2) such that s1 ++ s2 = s.
For example:

splits [1; 2; 3] = [ ([], [1; 2; 3]);
([1], [2; 3]);
([1; 2], [3]);
([1; 2; 3], []) ]

The function splits_many that is used in the Star case is a kind of n-way generalization of splits.
You don’t need to understand it in detail, but it enumerates all possible ways to split a list s into
one or more nonempty lists s1,...,sn such that s1 ++ ... ++ sn = s. For example:

splits_many [1; 2; 3] = [ [[1]; [2]; [3]];
[[1]; [2; 3]];
[[1; 2]; [3]];
[[1; 2; 3]] ]

(continued...)

Solution:

Fixpoint re_match (s : list nat) (re : reg_exp) : bool :=
match re with
| EmptySet => false
| EmptyStr => list_nat_eq s []
| Char n => list_nat_eq s [n]
| App re1 re2 =>

existsb (fun ’(s1, s2) => re_match s1 re1 && re_match s2 re2)
(splits s)

| Union re1 re2 => re_match s re1 || re_match s re2
| Star re’ =>

list_nat_eq s [] ||
existsb (fun ss’ => forallb (fun s’ => re_match s’ re’) ss’)

(splits_many s)
end.

For Reference

Numbers
Inductive nat : Type :=

| O : nat
| S : nat -> nat.

Fixpoint plus (n : nat) (m : nat) : nat :=
match n with

| O => m
| S n’ => S (plus n’ m)

end.

Notation "x + y" := (plus x y)(at level 50, left associativity) : nat_scope.

Fixpoint mult (n : nat) (m : nat) : nat :=
match n with

| 0 => 0
| S n’ => m + (mult n’ m)

end.

Notation "x * y" := (mult x y)(at level 40, left associativity) : nat_scope.

Fixpoint beq_nat (n m : nat) : bool :=
match n, m with
| O, O => true
| S n’, S m’ => beq_nat n’ m’
| _, _ => false
end.

Inductive le : nat -> nat -> Prop :=
| le_n : forall n, le n n
| le_S : forall n m, (le n m) -> (le n (S m)).

Notation "m <= n" := (le m n).

Inductive ev : nat -> Prop :=
| ev_0 : ev 0
| ev_SS : forall n : nat, ev n -> ev (S (S n)).

Binary Trees
Inductive tree (A : Type) :=

Empty : tree A
| Branch : A -> tree A -> tree A -> tree A.

Arguments Empty {A}.
Arguments Branch {A} _ _ _.

Lists
Inductive list (X:Type) : Type :=
| nil : list X
| cons : X -> list X -> list X.

Arguments nil {X}.
Arguments cons {X} _ _.
Notation "[ ]" := nil.
Notation "x :: l" := (cons x l) (at level 60, right associativity).
Notation "[ x ; .. ; y ]" := (cons x .. (cons y nil) ..).

Fixpoint app {X : Type} (l1 l2 : list X) : (list X) :=
match l1 with
| [] => l2
| h :: t => h :: (app t l2)
end.

Notation "x ++ y" := (app x y) (at level 60, right associativity).

Fixpoint repeat (n count : nat) : list nat :=
match count with
| O => nil
| S count’ => n :: (repeat n count’)
end.

Fixpoint In {A : Type} (x : A) (l : list A) : Prop :=
match l with
| [] => False
| x’ :: l’ => x’ = x \/ In x l’
end.

Fixpoint fold {X Y: Type} (f: X -> Y -> Y) (l: list X) (b: Y) : Y :=
match l with
| nil => b
| h :: t => f h (fold f t b)
end.

Fixpoint forallb {X : Type} (test : X -> bool) (l : list X) : bool :=
match l with
| [] => true
| x :: l’ => andb (test x) (forallb test l’)
end.

Fixpoint existsb {X : Type} (test : X -> bool) (l : list X) : bool :=
match l with

| [] => false
| x :: l’ => orb (test x) (existsb test l’)

end.

(* Equality on lists of nats *)
Fixpoint list_nat_eq (s1 s2 : list nat) : bool :=

match s1, s2 with
| n1 :: t1, n2 :: t2 => (n1 =? n2) && list_nat_eq t1 t2
| [], [] => true
| _, _ => false
end.

(* Enumerate all ways to split a list into two. *)
Fixpoint splits {X : Type} (s : list X) : list (list X * list X) :=

match s with
| [] => [([], [])]
| h :: t => ([], s) :: map (fun ’(s1, s2) => (h :: s1, s2)) (splits t)
end.

(* Enumerate all ways to split a list into _any number_ of _nonempty_ lists. *)
Fixpoint splits_many {X : Type} (s : list X) : list (list (list X)) :=

match s with
| [] => []
| h :: [] => [[s]]
| h :: t =>

map (fun t’ => [h] :: t’) (splits_many t) ++
map (fun t’ =>

match t’ with
| [] => t’
| u :: w => (h :: u) :: w
end) (splits_many t)

end.

Regular Expressions
Inductive reg_exp {T : Type} : Type :=
| EmptySet : reg_exp
| EmptyStr : reg_exp
| Char : T -> reg_exp
| App : reg_exp -> reg_exp -> reg_exp
| Union : reg_exp -> reg_exp -> reg_exp
| Star : reg_exp -> reg_exp.

Inductive exp_match {T: Type} : list T -> reg_exp -> Prop :=
| MEmpty : exp_match [] EmptyStr
| MChar : forall x, exp_match [x] (Char x)
| MApp : forall s1 re1 s2 re2,

exp_match s1 re1 ->
exp_match s2 re2 ->
exp_match (s1 ++ s2) (App re1 re2)

| MUnionL : forall s1 re1 re2,
exp_match s1 re1 ->
exp_match s1 (Union re1 re2)

| MUnionR : forall re1 s2 re2,
exp_match s2 re2 ->
exp_match s2 (Union re1 re2)

| MStar0 : forall re, exp_match [] (Star re)
| MStarApp : forall s1 s2 re,

exp_match s1 re ->
exp_match s2 (Star re) ->
exp_match (s1 ++ s2) (Star re).

Notation "s =~ re" := (exp_match s re) (at level 80).

